AD-A035 052 ROME AIR DEVELOPMENT CENTER GRIFFISS AFB N Y F/G 9/5

THE COMPLETE ELECTROMAGNETIC FIELDS IN THE FOCAL REGION OF A PA--ETC(U)

SEP 76 R L FANTE, R L TAYLOR

UNCLASSIFIED RADC-TR-76-295 NL

| OF |

AD
A035052

END

DATE
FILMED
3-77

The Complete Electromagnetic Fields in the Focal Region of a Paraboloidal Reflector

RONALD L. FANTE
RICHARD L. TAYLOR

ROME AIR DEVELOPMENT CENTER
AIR FORCE SYSTEMS COMMAND
GRIFFISS AIR FORCE BASE, NEW YORK 13441

ved: J. LEON POIRIER
Acting Chief
Microwave Detection Techniques Br.
Electromagnetic Sciences Division

ved: ALLAN C. SCHELL
Acting Chief
Electromagnetic Sciences Division

FOR THE COMMANDER:

Plans Office

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER RADC-TR-76-295	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) THE COMPLETE ELECTROMAGNETIC FIELDS IN THE FOCAL REGION OF A PARABOLOIDAL REFLECTOR		5. TYPE OF REPORT & PERIOD COVERED In-House
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Ronald L. Fante Richard L. Taylor		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Deputy for Electronic Technology (RADC/ETEP) Hanscom AFB, Massachusetts 01731		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 61102F 21530201
11. CONTROLLING OFFICE NAME AND ADDRESS Deputy for Electronic Technology (RADC/ETEP) Hanscom AFB, Massachusetts 01731		12. REPORT DATE Sep[redacted]76
		13. NUMBER OF PAGES 19
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office) 12 20p.		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE
16. DISTRIBUTION STATEMENT (of this Report) Approved for public release; distribution unlimited. 16 2153 17 02		
17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)		
18. SUPPLEMENTARY NOTES		
19. KEY WORDS (Continue on reverse side if necessary and identify by block number) Antennas Reflectors Electromagnetics		

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

By using the physical optics approximation we have developed a computer program to calculate the complete near field distribution of a paraboloid which is illuminated by a plane wave. This program is described and a number of results obtained with this program are presented, including some comparisons with known results for the fields in the focal plane.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

309050

1B

SECURITY CLASSIFICATION OF THIS PAGE(When Data Entered)

SECURITY CLASSIFICATION OF THIS PAGE(When Data Entered)

Contents

Illustrations

Illustrations

The Complete Electromagnetic Fields in the Focal Region of a Paraboloidal Reflector

1. INTRODUCTION

In designing optimum feed systems for Cassegrain reflector systems, it is highly desirable to have an accurate picture of the electromagnetic fields in the reflector focal region. In order to study these fields we have considered the case of a plane wave incident upon a large reflector, as shown in Figure 1, and have used the physical optics approximation to calculate the complete electromagnetic field distribution produced in the vicinity of the reflector focus.

2. THEORETICAL BACKGROUND

Let us consider a plane wave with electric and magnetic fields*

$$\underline{E}_i = E_o \hat{y} \exp [i (\omega t + kz)] , \tag{1a}$$

$$\underline{H}_i = H_o \hat{x} \exp [i(\omega t + kz)] , \tag{1b}$$

(Received for publication 29 September 1976)

*In Eq. (1) $\hat{x}$, $\hat{y}$, and $\hat{z}$ are unit vectors along x, y, and z. Also k is the wavenumber = $2\pi/\lambda$, where λ is the signal wavelength.

incident from the right upon the reflector in Figure 1. If we assume that the reflector surface is described by the arbitrary function

$$z = f(x, y) \quad , \tag{2}$$

it can be shown[1] that the magnetic field scattered by the reflector is given, in the physical optics approximation, by

$$\underline{H}_S = -\frac{H_o}{2\pi} \iint_{S_o} dxdy \left\{ \left(\hat{z} - \frac{\partial f}{\partial x} \hat{x} - \frac{\partial f}{\partial y} \hat{y} \right) \times \hat{x}\, e^{ikz} \right\} \times \nabla \left(\frac{e^{-ikR}}{R} \right), \tag{3}$$

where R is the distance from a source point (x, y, z) on the reflector to the field point (x_o, y_o, z_o), S_o is the projection of the reflector surface onto the x - y plane and x, y, and z are unit vectors.



Figure 1. Reflector Geometry

1. Silver, S. (1965) Microwave Antenna Theory and Design, Dover (New York).

If we assume that the reflector surface is a parabola with a focus at $z = F$, then Eq. (2) becomes

$$z = \frac{1}{4F}\left(x^2 + y^2\right) \quad , \tag{4}$$

and the projection of the reflector onto the x-y plane is a circle satisfying the equation

$$x^2 + y^2 = \left(\frac{D}{2}\right)^2 \quad , \tag{5}$$

where D is the diameter of the reflector. If we now use Eqs. (4) and (5) in (3) we obtain, after some manipulation

$$\underline{H}_S(x_o, y_o, z_o) = \frac{H_o}{2\pi}\int_{-D/2}^{D/2} dx \int_{-\gamma(x)}^{\gamma(x)} dy\, \phi(x, y)$$

$$\times \left\{ \hat{x}\left[z_o - \frac{x^2}{4F} + \frac{y^2}{4F} - \frac{y_o y}{2F}\right] + \hat{y}\,\frac{(x_o - x)y}{2F} - \hat{z}\,(x_o - x)\right\} \quad , \tag{6}$$

where

$$\gamma(x) = \left[\left(\frac{D}{2}\right)^2 - x^2\right]^{1/2} \quad ,$$

$$\phi(x, y) = \left(\frac{ik}{R^2} + \frac{1}{R^3}\right)\exp\left[-ik\left\{R - \frac{x^2}{4F} - \frac{y^2}{4F}\right\}\right] ,$$

$$R = \left[(x - x_o)^2 + (y - y_o)^2 + (z - z_o)^2\right]^{1/2} .$$

The electric field distribution can be obtained by employing the Maxwell equation

$$\nabla \times \underline{H}_S = i\omega\epsilon_o \underline{E}_S \quad . \tag{7}$$

The result for $\underline{E}_S$ is

$$E_x(x_o, y_o, z_o) = \frac{H_o}{2\pi i\omega\epsilon_o}\int_{-D/2}^{D/2} dx \int_{-\gamma(x)}^{\gamma(x)} dy\,(x_o - x)\left[y_o - y + \frac{y}{2F}z_o - \frac{y}{2F}\left(\frac{x^2}{4F} + \frac{y^2}{4F}\right)\right]\theta(x, y), \tag{8a}$$

$$E_y(x_o, y_o, z_o) = \frac{H_o}{\pi i\omega\epsilon_o} \int_{-D/2}^{D/2} dx \int_{-\gamma(x)}^{\gamma(x)} dy \quad \phi(x,y)$$

$$- \frac{H_o}{2\pi i\omega\epsilon_o} \int_{-D/2}^{D/2} dx \int_{-\gamma(x)}^{\gamma(x)} dy \quad \theta(x,y) \left\{ \alpha(x,y) \left[z_o - \frac{x^2}{4F} - \frac{y^2}{4F} \right] + (x_o - x)^2 \right\} , \tag{8b}$$

$$E_z(x_o, y_o, z_o) = \frac{H_o}{i\omega\epsilon_o \pi} \int_{-D/2}^{D/2} dx \int_{-\gamma(x)}^{\gamma(x)} dy \quad \frac{y}{2F} \phi(x,y)$$

$$- \frac{H_o}{2\pi i\omega\epsilon_o} \int_{-D/2}^{D/2} dx \int_{-\gamma(x)}^{\gamma(x)} dy \quad \theta(x,y) \left\{ \frac{y}{2F}(x_o - x)^2 - (y_o - y)\alpha(x,y) \right\} , \tag{8c}$$

where

$$\alpha(x,y) = z_o - \frac{x^2}{4F} + \frac{y^2}{4F} - \frac{y_o y}{2F} ,$$

$$\theta(x,y) = \left\{ \frac{3}{R^5} + \frac{i3k}{R^4} - \frac{k^2}{R^3} \right\} \exp \left[-ik \left(R - \frac{x^2}{4F} - \frac{y^2}{4F} \right) \right] .$$

Equations (6) and (8) are the formal expressions for the complete electromagnetic fields in the physical optics approximation. They represent a quite good approximation for the entire region $z > \frac{1}{4F}(x^2 + y^2)$, which is of interest to us. Of course, they are inaccurate for $z < 0$; over part of that region the geometrical theory of diffraction must be employed.

By observing Eqs. (6) and (8) it is clear that the scattered electric and magnetic fields possess certain symmetry properties. These are (for a fixed z_o)

$$H_x(x_o, -y_o) = H_x(x_o, y_o) , \tag{9a}$$

$$H_x(-x_o, y_o) = H_x(x_o, y_o) , \tag{9b}$$

$$H_y(x_o, -y_o) = -H_y(x_o, y_o) , \tag{9c}$$

$$H_y(-x_o, y_o) = -H_y(x_o, y_o) , \tag{9d}$$

$$H_z(x_o, -y_o) = H_z(x_o, y_o) \quad , \tag{9e}$$

$$H_z(-x_o, y_o) = -H_z(x_o, y_o) \quad . \tag{9f}$$

$$E_x(x_o, -y_o) = -E_x(x_o, y_o) \quad , \tag{10a}$$

$$E_x(-x_o, y_o) = -E_x(x_o, y_o) \quad , \tag{10b}$$

$$E_y(x_o, -y_o) = E_y(x_o, y_o) \quad , \tag{10c}$$

$$E_y(-x_o, y_o) = E_y(x_o, y_o) \quad , \tag{10d}$$

$$E_z(x_o, -y_o) = -E_z(x_o, y_o) \quad , \tag{10e}$$

$$E_z(-x_o, y_o) = E_z(x_o, y_o) \quad . \tag{10f}$$

Because of the aforementioned symmetry properties we have calculated $\underline{E}$ and $\underline{H}$ only for positive values of x_o and y_o; the values for negative x_o, y_o follow immediately from Eqs. (9) and (10).

3. RESULTS

We have developed a computer program to calculate the field components given by Eqs. (6) and (8). As an example of typical results of our program, we have studied a reflector such that

$$\frac{F}{D} = \frac{1}{3} \quad , \tag{11a}$$

$$\frac{D}{\lambda} = 60 \quad , \tag{11b}$$

where λ is the signal wavelength, and have calculated the field distribution in the planes $z_o = 0.95F$, $0.967F$, $0.983F$, and $1.0F$. In Figures 2 and 3 we show* the amplitude and phase of the electric and magnetic fields in the plane $z_o = 0.95F$. The fields shown are those along the line $y_o = 0$, for differing values of x_o. In Figures 4 and 5 we show the fields along the line $x_o = 0$ for differing values of y_o.

*In all the results of Figures 2 to 11 we have assumed $H_o = 2\pi$.



Figure 2. Magnitude of the Fields Along the y_o = 0 Axis for z = 0.95F



Figure 3. Phase of the Fields Along the y_o = 0 Axis for z_o = 0.95F



Figure 4. Magnitude of the Fields Along the x_o = 0 Axis for z_o = 0.95F



Figure 5. Phase of the Fields Along the x_o = 0 Axis for z_o = 0.95F

We note, upon comparing Figures 2 and 4, that if the results are known along the $y_o = 0$ axis we can immediately obtain those along the axis $x_o = 0$ by replacing $|H_x|$ by $|E_y|/Z_o$, $|H_z|$ by $|E_z|/Z_o$, and $|E_y|/Z_o$ by $|H_x|$, where Z_o is the impedance of vacuum. Because of this duality, in the remaining figures we will only show results along either the $x_o = 0$ or the $y_o = 0$ axis. In Figures 6 and 7 we show the fields in the plane $z_o = 0.967F$; in Figures 8 and 9 we show the results in the plane $z_o = 0.983F$, and finally, in Figures 10 and 11 we show the results in the focal plane. The components E_x and H_y are now shown bec_use they are both zero.

There are several observations which should be made regarding our results:

(1) The results of Figure 10 for the transverse fields in the focal plane agree with those calculated earlier by Minnett et al.[2] (see Figure 11 of Minett's paper, for our case $\theta_o \sim 74^o$).

(2) The cross-polarized fields $|E_z|/Z_o$ and $|H_z|$ are generally of the same order as $|H_x|$ and $|E_y|/Z_o$, except very near to the z_o axis. This is true even in the focal plane, and even holds within the focal spot (that is, we call the transverse dimension of the first null in Figure 10 the focal spot size, and this is of order $\lambda F/D$) as can be seen from Figure 10, where $|E_z|/Z_o$ is small near the center of the focal spot (y_o near zero) but is large near the outer edge ($y_o \simeq 0.5\lambda$).

(3) In Figures 2 to 11 we have shown the fields on the x and y axes where $E_x = H_y = 0$. This should not imply that $E_x = H_y = 0$ off these axes. In Figure 12 we show the field distribution along an axis (see Figure 13) oriented 45^o relative to the x axis in the plane z_o. Note that both E_x and H_y are nonzero, although they are considerably smaller than the other field components.

---

2. Minnett, H.C. and Thomas, B. (1968) Fields in the image plane of symmetrical focusing reflectors. Proc. IEEE, 115:1419-1430.



Figure 6. Magnitude of the Fields Along the $y_0 = 0$ Axis for $z_0 = 0.967F$



Figure 7. Phase of the Fields Along the $y_0 = 0$ Axis for $z_0 = 0.967F$



Figure 8. Magnitude of the Fields Along the $y_0 = 0$ Axis for $z_0 = 0.983F$



Figure 9. Phase of the Fields Along the $y_0 = 0$ Axis for $z_0 = 0.983F$





Figure 10. Magnitude of the Fields Along the $x_o = 0$ Axis for $z_o = F$ (focal plane)

Figure 11. Phase of the Fields Along the $x_o = 0$ Axis for $z_o = F$ (focal plane)



Figure 12. Magnitude of the Fields Along an Axis Oriented at 45° with Respect to the x Axis, for $z_o = 0.967F$



Figure 13. Geometry for the Results in Figure 12

Appendix A

In this appendix we present a Fortran listing of the computer program used to calculate $\underline{E}$ and $\underline{H}$. Note that the quantities printed out for $\underline{E}$ are actually the electric field normalized by Z_o rather than $\underline{E}$. The inputs to the program are:

D = diameter of reflector,

F = focal length,

K = *wavenumber* = $2\pi/\lambda$,

X0, Y0, Z0 = coordinates of observation point,

XTOL = YTOL usually set to 10^{-3}.

```
      PROGRAM MAIN(INPUT, OUTPUT)
      IMPLICIT  COMPLEX(Q)
      COMMON/ONE/X0, ONEO2F, ONEO4F, Z0, K, Y0, DOV2SQ, QCONST
      COMMON/QXI/QXINTEG(6)
      COMMON/XTOL/XTOL
      COMMON/YTOL/YTOL
      REAL  K
      NAMELIST/XINPUT/ D, F, K, Z0, X0, Y0, XTOL, YTOL
      READ XINPUT
      QCONST = (1.0, 0.0) / CMPLX(0.0, K)
      ONEO2F = 0.5 / F
      ONEO4F = 0.25 / F
      DOV2SQ = (D / 2.0)**2
      XB = D / 2.0
      XA = -XB
      CALL XINTEG(XB, XA)
      PRINT 600,  QXINTEG
      STOP
  600 FORMAT(6H HX = ,1P2E12.5/6H HY = ,2E12.5/6H HZ = ,2E12.5/
     1      6H EX = ,2E12.5/6H EY = ,2E12.5/6H EZ = ,2E12.5)
      END
```

```
      SUBROUTINE XINTEG(XB, XA)
      IMPLICIT  COMPLEX(Q)
      COMMON/QXI/QXINTEG(6)
      COMMON/QYI/QYINTEG(6)
      COMMON/XTOL/TOL
      DIMENSION  QTWO(6), QFOUR(6), QENDS(6), QTOTAL(6)
      H = (XB - XA) / 2.0
      N = 1
      CALL YINTEG(H+XA)
      DO 10  J=1,6
      QTWO(J) = (0.0, 0.0)
   10 QFOUR(J) = QYINTEG(J)
      CALL YINTEG(XA)
      DO 20  J=1,6
   20 QENDS(J) = QYINTEG(J)
      CALL YINTEG(XB)
      DO 30  J=1,6
      QENDS(J) = QENDS(J) + QYINTEG(J)
   30 QTOTAL(J) = (QENDS(J) + 4.0*QFOUR(J) ) * H / 3.0
   40 DO 50  J=1,6
   50 QXINTEG(J) = QTOTAL(J)
      Y = H = H / 2.0
      N = 2 * N
      DO 60  J=1,6
      QTWO(J) = QTWO(J) + QFOUR(J)
   60 QFOUR(J) = (0.0, 0.0)
      I = 0
   70 I = I + 1
      CALL YINTEG(Y+XA)
      DO 80  J=1,6
   80 QFOUR(J) = QFOUR(J) + QYINTEG(J)
      Y = Y + H + H
      IF(I .LT. N)  GO TO 70
      IFLAG = 0
      PRINT *
      DO 90  J=1,6
      QTOTAL(J) = (QENDS(J) + 2.0*QTWO(J) + 4.0*QFOUR(J) ) * H / 3.0
      QDENOM = QTOTAL(J)
      IF(CABS(QDENOM) .LT. TOL)  QDENOM = CMPLX(TOL, 0.0)
      IF(CABS( (QXINTEG(J) - QTOTAL(J) )/QDENOM) .GT. TOL)  IFLAG = 1
      PRINT *,  CABS( QTOTAL(J) )
   90 CONTINUE
      IF(IFLAG .EQ. 1)  GO TO 40
      DO 100  J=1,6
  100 QXINTEG(J) = QTOTAL(J)
      RETURN
      END
```

```
      SUBROUTINE YINTEG(X)
      IMPLICIT  COMPLEX(Q)
      COMMON/ONE/X0, ONEO2F, ONEO4F, Z0, K, Y0, DOV2SQ, QCONST
      COMMON/QQQ/QINT(6)
      COMMON/QYI/QYINTEG(6)
      COMMON/XXX/XDIFF, XDIFF2, XSQ, EX
      COMMON/YTOL/TOL
      REAL  K
      DIMENSION  QTWO(6), QFOUR(6), QENDS(6), QTOTAL(6)
      EX = X
      XSQ = X**2
      XDIFF = X0 - X
      XDIFF2 = XDIFF**2
      YB = GAMX = SQRT(DOV2SQ - XSQ)
      YA = -GAMX
      H = GAMX
      N = 1
      CALL FIELDS(H+YA)
      DO 10  J=1,6
      QTWO(J) = (0.0, 0.0)
   10 QFOUR(J) = QINT(J)
      CALL FIELDS(YA)
      DO 20  J=1,6
   20 QENDS(J) = QINT(J)
      CALL FIELDS(YB)
      DO 30  J=1,6
      QENDS(J) = QENDS(J) + QINT(J)
   30 QTOTAL(J) = (QENDS(J) + 4.0*QFOUR(J) ) * H / 3.0
   40 DO 50  J=1,6
   50 QYINTEG(J) = QTOTAL(J)
      Y = H = H / 2.0
      N = 2 * N
      DO 60  J=1,6
      QTWO(J) = QTWO(J) + QFOUR(J)
   60 QFOUR(J) = (0.0, 0.0)
      I = 0
   70 I = I + 1
      CALL FIELDS(Y+YA)
      DO 80  J=1,6
   80 QFOUR(J) = QFOUR(J) + QINT(J)
      Y = Y + H + H
      IF(I .LT. N)  GO TO 70
      IFLAG = 0
      DO 90  J=1,6
      QTOTAL(J) = (QENDS(J) + 2.0*QTWO(J) + 4.0*QFOUR(J) ) * H / 3.0
      QDENOM = QTOTAL(J)
      IF(CABS(QDENOM) .LT. TOL)  QDENOM = CMPLX(TOL, 0.0)
      IF(CABS( (QYINTEG(J) - QTOTAL(J) )/QDENOM) .GT. TOL)  IFLAG = 1
   90 CONTINUE
      IF(IFLAG .EQ. 1)  GO TO 40
      DO 100  J=1,6
  100 QYINTEG(J) = QTOTAL(J)
      RETURN
      END
```

```
      SUBROUTINE FIELDS(Y)
      IMPLICIT COMPLEX(Q)
      COMMON/ONE/X0, ONEO2F, ONEO4F, Z0, K, Y0, DOV2SQ, QCONST
      COMMON/QQQ/QINT(6)
      COMMON/XXX/XDIFF, XDIFF2, XSQ, X
      REAL  K
      YSQ = Y**2
      YDIFF = Y0 - Y
      YDIFF2 = YDIFF**2
      TERM1 = Z0 - ONEO4F*(XSQ + YSQ)
      RSQ = XDIFF2 + YDIFF2 + TERM1**2
      R   = SQRT(RSQ)
      ONEOR2 = 1.0 / RSQ
      ONEOR  = 1.0 / R
      QCEXP  = CEXP(CMPLX(0.0, -K*(R + TERM1 - Z0) ) )
      QPHI   = QCEXP * CMPLX(ONEOR, K) * ONEOR2
      QTHETA = QCEXP * CMPLX(3.0*ONEOR2-K**2, 3.0*K*ONEOR)*ONEOR2*ONEOR
      TERM2  = Z0 - ONEO4F * (XSQ - Y*(Y - 2.0*Y0) )
      QINT(1) = QPHI * TERM2
      QINT(3) = -QPHI * XDIFF
      QINT(2) = -QINT(3) * Y * ONEO2F
      QINT(4) = XDIFF * (YDIFF + ONEO2F*Y*TERM1) * QTHETA * QCONST
      QINT(5) = (2.0*QPHI - QTHETA*(TERM2*TERM1 + XDIFF2) )*QCONST
      QINT(6) = (2.0*QPHI*Y*ONEO2F - QTHETA*(XDIFF2*Y*ONEO2F - YDIFF*
     1      TERM2) )*QCONST
      RETURN
      END
```

MISSION
of
Rome Air Development Center

RADC plans and conducts research, exploratory and advanced development programs in command, control, and communications (C^3) activities, and in the C^3 areas of information sciences and intelligence. The principal technical mission areas are communications, electromagnetic guidance and control, surveillance of ground and aerospace objects, intelligence data collection and handling, information system technology, ionospheric propagation, solid state sciences, microwave physics and electronic reliability, maintainability and compatibility.